Abstract

Contents

1 Introduction

2 Agda proof assistant

Agda is a dependently typed programming language based on intuitionistic type theory. By encoding mathematical propositions as types and their proofs as programs, we can ensure that our reasoning is correct and consistent. Agda's type system also provides powerful tools for automatically checking the correctness of proofs[2].

2.1 Propositions as types

Propositions as types associates logical propositions with types in a programming language. It is based on the idea that a proof of a proposition is analogous to a program that satisfies the type associated with the proposition.

In this context, the introduction and elimination rules for logical connectives can be seen as operations that construct and deconstruct values of the corresponding types. For example, the introduction rule for conjunction says that if we have proofs of two propositions, we can construct a proof of their conjunction by pairing the two proofs together. This can be seen as a function that takes two values of the corresponding types and returns a pair value.

On the other hand, the elimination rule for conjunction says that if we have a proof of a conjunction, we can extract proofs of its two conjuncts by projecting the pair onto each component. This can be seen as a function that takes a pair value and returns two values of the corresponding types.

This is similar to the concept of product types in programming languages, where a product type is a type that represents a pair of values. The introduction form of a product type is a pair, and the elimination forms are projection functions that extract the individual components of the tuple. This chart summarizes the correspondence between proposition and types and between proofs and programs.

Prop	Type
$\top$	unit
$\bot$	void
$\phi_1 \wedge \phi_2$	$\tau_1 \times \tau_2$
$\phi_1 \supset \phi_2$	$\tau_1 \to \tau_2$
$\phi_1 \vee \phi_2$	$\tau_1 + \tau_2$

Table 1: Propositions as types

This allows us to reason about logical propositions in terms of programming language types, and to use the tools and techniques of programming languages like `Agda` to reason about logical proofs.

2.2 Simply typed functions and datatypes

A data declaration is used to introduce datatypes, including their name, type, and constructors along with their types. An example of this is the declaration of the boolean type:

```
data Bool : Type where
  true : Bool
  false : Bool
```

This states that `Bool` is a data type with `true` and `false` as constructors. Functions over this datatype `Bool` can be defined using pattern matching, similar to Haskell. For instance we can define a function `not` for `Bool` as follows:

```
not : Bool → Bool
not true = false
not false = true
```

We start by defining the type of not as a function from `Bool` to `Bool` and then we define the function by using pattern matching on the arguments. Agda checks that the pattern covers all cases and will not accept a function with missing patterns.

The natural numbers can be defined as the datatype:

```
data ℕ : Type where
  zero : ℕ
  suc : ℕ → ℕ
```

A natural number is either zero or a successor of another natural number. This is called an *inductively defined type*. We can define addition on the natural numbers with a recursive function.

```
_+_ : ℕ → ℕ → ℕ
zero + m = m
suc n + m = suc (n + m)
```

If a name contains underscores (_) in the definition, the underscores represent where the arguments go. So in this case we get an infix operator and we write m + n instead of + m n, which would have been the case if the name was just +. We can set the precedence of an infix operator with an `infix` declaration:

```
infix 25 _+_
```

Datatypes can also be parameterized by other types. The type of lists with elements of an arbitrary type is defined as:

```
infix 20 _::_
data List (A : Type) : Type where
  [] : List A
  _::_ : A → List A → List A
```

2.3 Dependent types

A dependent type is a type that depends on elements of another type. An example of a dependent type is a dependent function, where the result type depends on the value of the argument. In Agda, this is denoted by (x : A) → B, representing functions that take an argument x of type A and produce a result of type B. A special case is when x itself is a type. For instance, we can define the identity function

```
id : (A : Type) → A → A
id A x = x
```

This function takes a type argument A and an element x of type A, and returns x. In `Agda` it is possible to use implicit arguments. To declare an argument as implicit we use curly braces instead of parenthesis when declaring the type argument. In particular, {A : Set} → B means the same thing as (A : Set) → B, but we don't need to provide the type explicitly, the type checker will try to infer it for us. We can now redefine the identity function above as follows:

```
id' : {A : Type} → A → A
id' x = x
```

Note that we no longer need to supply the type when the function is applied.

2.4 Cubical Agda

In this project we will be dealing with quotients, and for that we need to use `Cubical Agda`. This extends `Agda` with features from Cubical Type Theory, which is needed when we deal with quotients.[1] We use only a small part of the `agda/cubical` library, and any cubical theory is beyond this thesis.

3 Propositional calculus in Agda

Propositional calculus is a formal system that consists of a set of propositional constants, symbols, inference rules, and axioms. The symbols in propositional calculus represent logical connectives and parentheses, and are used to construct well-formed formulas that follow the syntax of the system. The inference rules of propositional calculus specify how these symbols can be used to derive additional statements from the initial assumptions, which are given by the axioms of the system.

The semantics of propositional calculus define how the expressions in the system correspond to truth values, typically "true" or "false".

3.1 Formulas

Definition 3.1 (Language)**.** *The language $\mathcal{L}$ of propositional calculus consists of*

- *proposition symbols:* $p_0, p_1, \ldots, p_n$*,*
- *logical connectives:* $\wedge, \vee, \neg, \top, \bot$*,*
- *auxiliary symbols: (,).*

Note that we have omitted the common logical connectives $\rightarrow$ and $\leftrightarrow$. This is becuase we can define them using other connectives,

$$\phi \rightarrow \psi \stackrel{\text{def}}{=} \neg\phi \vee \psi,$$
$$\phi \leftrightarrow \psi \stackrel{\text{def}}{=} (\neg\phi \vee \psi) \wedge (\neg\psi \vee \phi),$$

making them reduntant. It is possible to choose an even smaller set of connectives [3], but we choose this as it is convenient.

Definition 3.2 (Well formed formula)**.** *The set of well formed formulas is inductively defined as*

- *any propositional constant* $p_0, p_1, \ldots, p_n$ *is a well formed formula,*
- $\top$ *and* $\bot$ *are well formed formulas,*
- *if* p *is a well formed formula, then so is*
$$\neg p,$$
- *if* p_i *and* p_j *are well formed formulas, then so are*
$$p_i \wedge p_j \quad \text{and} \quad p_i \vee p_j.$$

The formula $\top$ *should be thought of as the proposition that is always true, and the formula* $\bot$ *interpreted as the proposition that is always false.*

We represent the concept of a well formed formula in Agda as a data type.

```
data Formula : Type where
  _∧_ : Formula → Formula → Formula
  _∨_ : Formula → Formula → Formula
  ¬_  : Formula → Formula
  const : ℕ → Formula
  ⊥   : Formula
  ⊤   : Formula
```

3.2 Context

Definition 3.3 (Context)**.** *A set of sentences in the language* $\mathcal{L}$*. The set is defined inductively as*

- *the empty set is a context*

- *if Γ is a context, then $\Gamma \cup \{\phi\}$ is also a context, where ϕ a formula.*

In Agda we can define a data type for context.

```
data ctxt : Type where
  ∅  : ctxt
  _:_ : ctxt → Formula → ctxt
```

We also need a way to determine if a given formula is in a given context.

Definition 3.4 (Lookup)**.** *For all contexts Γ and all formulas ϕ and ψ*

- $\phi \in \Gamma \cup \{\phi\}$,
- *if $\phi \in \Gamma$, then $\phi \in \Gamma \cup \{\psi\}$.*

We represent this as a data type in Agda

```
data _∈_ : Formula → ctxt → Type where
  Z : ∀ {Γ φ} → φ ∈ Γ : φ
  S : ∀ {Γ φ ψ} → φ ∈ Γ → φ ∈ Γ : ψ
```

3.3 Inference rules

For the inference rules we introduce a data type for provability

```
data _⊢_ : ctxt → Formula → Type where

                    ⋮
```

where we will define our inference rules on the following form:

```
rulename : { ... : ctxt} { ... : Formula}
        -> premise.1
        -> premise.2
              :
        -> premise.n
        -> conclusion
```

3.3.1 Law of excluded middle

Definition 3.5. *The law of excluded middle states that for every proposition, either the proposition or its negation is true.*

$$\frac{}{\Gamma \vdash \phi \lor \neg\phi}\ \text{LEM}$$

The law of excluded middle in Agda:

```
LEM : {Γ : ctxt} {φ : Formula}
    → Γ ⊢ φ ∨ ¬ φ
```

3.3.2 Logical connectives

Rules for the logical connectives come in pairs of introduction and elimination rules, a part from $\top$, which has only an introduction rule, and $\bot$, which has only an elimination rule. These rules are based on [3].

The introductory rule for conjunction states that if there is a derivation of ϕ in the context Γ, and a derivation of ψ in the context Γ, then we can conclude that there is a derivation of $\phi \wedge \psi$ in Γ.

$$\frac{\Gamma \vdash \phi \qquad \Gamma \vdash \psi}{\Gamma \vdash \phi \wedge \psi}\ \wedge\text{-I}$$

The conjunction introduction rule in `Agda`:

```
∧-I : {Γ : ctxt} {φ ψ : Formula}
    → Γ ⊢ φ
    → Γ ⊢ ψ
    → Γ ⊢ φ ∧ ψ
```

The accompanying elimination rules says that if there is some derivation concluding in $\phi \wedge \psi$ in Γ, then we can conclude that there is a derivation of ϕ and a derivation of ψ in Γ.

$$\frac{\Gamma \vdash \phi \wedge \psi}{\Gamma \vdash \phi}\ \wedge\text{-E}_1 \qquad \frac{\Gamma \vdash \phi \wedge \psi}{\Gamma \vdash \psi}\ \wedge\text{-E}_2$$

The conjunction elimination rules in `Agda`:

```
∧-E₁ : {Γ : ctxt} {φ ψ : Formula}
     → Γ ⊢ φ ∧ ψ
     → Γ ⊢ φ


∧-E₂ : {Γ : ctxt} {φ ψ : Formula}
     → Γ ⊢ φ ∧ ψ
     → Γ ⊢ ψ
```

For disjunction we have two introductory rules. If Γ proves some formula ψ, then Γ proves $\phi \vee \psi$. In the same way Γ proves $\phi \vee \psi$ if Γ proves ϕ.

$$\frac{\Gamma \vdash \psi}{\Gamma \vdash \phi \vee \psi}\ \vee\text{-I}_1 \qquad \frac{\Gamma \vdash \phi}{\Gamma \vdash \phi \vee \psi}\ \vee\text{-I}_2$$

The disjunction introduction rules in `Agda`:

```
∨-I₁ : {Γ : ctxt} {φ ψ : Formula}
     → Γ ⊢ ψ
     → Γ ⊢ φ ∨ ψ
```

$$\begin{aligned}\lor\text{-}\mathsf{I}_2 :\ & \{\Gamma : \mathsf{ctxt}\}\ \{\phi\ \psi : \mathsf{Formula}\} \\ & \to \Gamma \vdash \phi \\ & \to \Gamma \vdash \phi \lor \psi\end{aligned}$$

The eliminination rule for disjunction is a bit more complicated. If Γ proves $\phi \lor \psi$, then we can conclude Γ proves γ if the extended contexts Γ, ϕ and Γ, ψ both prove γ.

$$\frac{\Gamma \vdash \phi \lor \psi \qquad \Gamma, \phi \vdash \gamma \qquad \Gamma, \psi \vdash \gamma}{\Gamma \vdash \gamma}\ \lor\text{-E}$$

The disjunction elimination rule in `Agda`:

$$\begin{aligned}\lor\text{-E} :\ & \{\Gamma : \mathsf{ctxt}\}\ \{\phi\ \psi\ \gamma : \mathsf{Formula}\} \\ & \to \Gamma \vdash \phi \lor \psi \\ & \to \Gamma : \phi \vdash \gamma \\ & \to \Gamma : \psi \vdash \gamma \\ & \to \Gamma \vdash \gamma\end{aligned}$$

Definition 3.6. *A context Γ is inconsistent if $\Gamma \vdash \bot$. A context that is not inconsistent, i.e. $\Gamma \nvdash \bot$, is called consistent.*

This definition and the law of excluded middle together motivates the introduction and elimination rules for negation.

$$\frac{\Gamma, \phi \vdash \bot}{\Gamma \vdash \neg\phi}\ \neg\text{-I} \qquad\qquad \frac{\Gamma \vdash \phi \qquad \Gamma \vdash \neg\phi}{\Gamma \vdash \bot}\ \neg\text{-E}$$

The negation rules in `Agda`:

$$\begin{aligned}\neg\text{-I} :\ & \{\Gamma : \mathsf{ctxt}\}\ \{\phi : \mathsf{Formula}\} \\ & \to \Gamma : \phi \vdash \bot \\ & \to \Gamma \vdash \neg\ \phi\end{aligned}$$

$$\begin{aligned}\neg\text{-E} :\ & \{\Gamma : \mathsf{ctxt}\}\ \{\phi : \mathsf{Formula}\} \\ & \to \Gamma \vdash \phi \\ & \to \Gamma \vdash \neg\ \phi \\ & \to \Gamma \vdash \bot\end{aligned}$$

Since $\top$ is always trivially true it can be introduced with no premise.

$$\frac{}{\Gamma \vdash \top}\ \top\,\mathrm{I}$$

The $\top$ introduction rule in `Agda`:

$$\begin{aligned}\top\text{-I} :\ & \{\Gamma : \mathsf{ctxt}\} \\ & \to \Gamma \vdash \top\end{aligned}$$

If a context is inconsistent one can derive anything from it, which leads to the elimination rule for $\bot$.

$$\frac{\Gamma \vdash \bot}{\Gamma \vdash \phi}\ \bot\text{-E}$$

The $\bot$ elimination rule in `Agda`:

```
⊥-E : {Γ : ctxt} {ϕ : Formula}
    → Γ ⊢ ⊥
    → Γ ⊢ ϕ
```

3.3.3 Structural rules

Weakening is a structural rule that states that we can extend the hypothesis with additional members,

$$\frac{\Gamma \vdash \phi}{\Gamma, \psi \vdash \phi}\ \text{WEAKENING}$$

The weakening rule in `Agda`:

```
weakening : {Γ : ctxt} {ϕ ψ : Formula}
          → Γ ⊢ ψ
          → Γ ∷ ϕ ⊢ ψ
```

The structural rule exchange lets us permute the formulas in the context. We will use a stricter version of exchange, where we can permute the two formulas at the end only. This is easier to implement and still suits our needs.

$$\frac{\Gamma, \phi, \psi \vdash \gamma}{\Gamma, \psi, \phi \vdash \gamma}\ \text{EXCHANGE}$$

The exchange rule in `Agda`:

```
exchange : {Γ : ctxt} {ϕ ψ γ : Formula}
         → (Γ ∷ ϕ) ∷ ψ ⊢ γ
         → (Γ ∷ ψ) ∷ ϕ ⊢ γ
```

3.4 Properties of a propositional calculus

We prove some properties of propositional calculus

3.4.1 Equivalence relation

Definition 3.7. *Let S be the set of all the sentences of $\mathcal{L}$. Define the relation $\sim$ such that for $\phi, \psi \in S$,*

$$\phi \sim \psi \qquad \textit{iff} \qquad \Gamma, \phi \vdash \psi \textit{ and } \Gamma, \psi \vdash \phi$$

This is represented in `Agda` as a product type.

```
_∼_ : Formula → Formula → Type
φ ∼ ψ = Γ ∶ φ ⊢ ψ × Γ ∶ ψ ⊢ φ
```

Before we prove that this is an equivalence relation we will prove a lemma.

Lemma 3.1. *Given $\Gamma, \phi \vdash \psi$ and $\Gamma, \psi \vdash \gamma$, it follows that $\Gamma, \phi \vdash \gamma$*

Proof. This is done through natural deduction

$$\cfrac{\cfrac{\Gamma, \phi \vdash \psi}{\Gamma, \phi \vdash \psi \lor \gamma}\ \scriptstyle{\vee\text{-I}_2} \qquad \cfrac{\cfrac{\Gamma, \psi \vdash \gamma}{\Gamma, \psi, \phi \vdash \gamma}\ \scriptstyle{\text{WEAKENING}}}{\Gamma, \phi, \psi \vdash \gamma}\ \scriptstyle{\text{EXCHANGE}} \qquad \cfrac{\gamma \in \Gamma, \phi, \gamma}{\Gamma, \phi, \gamma \vdash \gamma}\ \scriptstyle{\text{AXIOM}}}{\Gamma, \phi \vdash \gamma}\ \scriptstyle{\vee\text{-E}}$$

□

And the corresponding `Agda` proof of the lemma:

```
⊢trans : ∀ {φ ψ γ : Formula} → Γ ∶ φ ⊢ γ → Γ ∶ γ ⊢ ψ → Γ ∶ φ ⊢ ψ
⊢trans A B = ∨-E (∨-I₂ A) (exchange (weakening B)) (axiom Z)
```

Theorem 3.1. *The relation $\sim$ is an equivalence relation.*

Proof. Reflexivity follows from the axiom rule and definition 3.4.

$$\cfrac{\phi \in \Gamma, \phi}{\Gamma, \phi \vdash \phi}\ \scriptstyle{\text{AXIOM}}$$

This gives the `Agda` proof:

```
∼-refl : ∀ (φ : Formula) → φ ∼ φ
∼-refl _ = axiom Z , (axiom Z)
```

It should be clear that $\Gamma, \phi \vdash \psi$ and $\Gamma, \psi \vdash \phi$ is just a pair of proofs, hence it does not matter in which order we give them. This means that the relation is also symmetric.

```
∼-sym : ∀ {φ ψ : Formula} → φ ∼ ψ → ψ ∼ φ
∼-sym (A , B) = B , A
```

For transitivity we need to prove that, given $\phi \sim \gamma$ and $\gamma \sim \psi$, it holds that $\phi \sim \psi$. By definition 3.7 we have proof of

(i) $\Gamma, \phi \vdash \gamma$,

(ii) $\Gamma, \gamma \vdash \phi$,

(iii) $\Gamma, \gamma \vdash \psi$, and

(iv) $\Gamma, \psi \vdash \gamma$.

Now we can apply Lemma 3.1 on (i) and (iii) to get $\Gamma, \phi \vdash \psi$, and again to (iv) and (ii) to get $\Gamma, \psi \vdash \phi$. The proof in Agda is straight forward.

$$\begin{array}{l}
\sim\text{-trans} : \forall \{\phi\ \psi\ \gamma : \mathsf{Formula}\} \to \phi \sim \gamma \to \gamma \sim \psi \to \phi \sim \psi \\
\sim\text{-trans}\ x\ y = \vdash\!\mathsf{trans}\ (\mathsf{proj}_1\ x)\ (\mathsf{proj}_1\ y)\ ,\ \vdash\!\mathsf{trans}\ (\mathsf{proj}_2\ y)\ (\mathsf{proj}_2\ x)
\end{array}$$

□

3.4.2 Commutativity

Proposition 1. $\phi \wedge \psi \sim \psi \wedge \phi$

Proof. We need to show $\Gamma, \phi \wedge \psi \vdash \psi \wedge \phi$ and $\Gamma, \psi \wedge \phi \vdash \phi \wedge \psi$. To show $\Gamma, \phi \wedge \psi \vdash \psi \wedge \phi$, we have:

$$\dfrac{\dfrac{\dfrac{\phi \wedge \psi \in \Gamma, \phi \wedge \psi}{\Gamma, \phi \wedge \psi \vdash \phi \wedge \psi}\,\textsc{axiom}}{\Gamma, \phi \wedge \psi \vdash \psi}\,\wedge\text{-E}_2 \qquad \dfrac{\dfrac{\phi \wedge \psi \in \Gamma, \phi \wedge \psi}{\Gamma, \phi \wedge \psi \vdash \phi \wedge \psi}\,\textsc{axiom}}{\Gamma, \phi \wedge \psi \vdash \phi}\,\wedge\text{-E}_1}{\Gamma \vdash \psi \wedge \phi}\,\wedge\text{-I}$$

The other proof is identical up to renaming of the formulas, so we omitt it. Together they prove $\phi \wedge \psi \sim \psi \wedge \phi$. This corresponds to the Agda proof:

$$\begin{array}{l}
\wedge\text{-comm} : \forall \{\phi\ \psi : \mathsf{Formula}\} \to \Gamma : \phi \wedge \psi \vdash \psi \wedge \phi \\
\wedge\text{-comm} = \wedge\text{-I}\ (\wedge\text{-E}_2\ (\mathsf{axiom\ Z}))\ (\wedge\text{-E}_1\ (\mathsf{axiom\ Z}))
\end{array}$$

Thus, we have shown that the two statements are equivalent in Agda:

$$\begin{array}{l}
\mathsf{comm\text{-}eq}\text{-}\wedge : \forall \{\phi\ \psi : \mathsf{Formula}\} \to \phi \wedge \psi \sim \psi \wedge \phi \\
\mathsf{comm\text{-}eq}\text{-}\wedge = \wedge\text{-comm}\ ,\ \wedge\text{-comm}
\end{array}$$

□

Proposition 2. $\phi \vee \psi \sim \psi \vee \phi$

Proof. We need to show $\Gamma, \phi \vee \psi \vdash \psi \vee \phi$ and $\Gamma, \psi \vee \phi \vdash \phi \vee \psi$. Up to renaming of the formulas, the proofs are identical, so it suffices to show $\Gamma, \phi \vee \psi \vdash \psi \vee \phi$. Let $\Gamma' = \Gamma \cup \{\phi \vee \psi\}$, then by natural deduction we can show $\Gamma' \vdash \psi \vee \phi$.

$$\dfrac{\dfrac{\phi \vee \psi \in \Gamma'}{\Gamma' \vdash \phi \vee \psi}\,\textsc{axiom} \qquad \dfrac{\dfrac{\phi \in \Gamma', \phi}{\Gamma', \phi \vdash \phi}\,\textsc{axiom}}{\Gamma', \phi \vdash \psi \vee \phi}\,\vee\text{-I}_1 \qquad \dfrac{\dfrac{\psi \in \Gamma', \psi}{\Gamma', \psi \vdash \psi}\,\textsc{axiom}}{\Gamma', \psi \vdash \psi \vee \phi}\,\vee\text{-I}_2}{\Gamma' \vdash \psi \vee \phi}\,\vee\text{-E}$$

Therefore, we have shown $\Gamma, \phi \vee \psi \vdash \psi \vee \phi$ and $\Gamma, \psi \vee \phi \vdash \phi \vee \psi$. Hence, $\phi \vee \psi \sim \psi \vee \phi$. In Agda we can write the corresponding proof as follows:

```
∨-comm : {φ ψ : Formula} → Γ : φ ∨ ψ ⊢ ψ ∨ φ
∨-comm = ∨-E (axiom Z) (∨-I₁ (axiom Z)) (∨-I₂ (axiom Z))
```

Then, we show that the equivalence of the statements hold in Agda:

```
comm-eq-∨ : ∀ {φ ψ : Formula} → φ ∨ ψ ~ ψ ∨ φ
comm-eq-∨ = ∨-comm , ∨-comm
```

□

3.4.3 Associativity

Proposition 3. $\phi \wedge (\psi \wedge \gamma) \sim (\phi \wedge \psi) \wedge \gamma$

Proof. We need to show $\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash (\phi \wedge \psi) \wedge \gamma$ and $\Gamma, (\phi \wedge \psi) \wedge \gamma \vdash \phi \wedge (\psi \wedge \gamma)$. Using natural deduction, we can prove both statements. Since the two proofs are very similar, we will only present the first one. However, the complete deduction tree is too large to include here, but can be found in A.1 for reference. In Agda, the corresponding proof of the first entailment is:

```
∧-assoc1 : ∀ {φ ψ γ : Formula} → Γ : φ ∧ (ψ ∧ γ) ⊢ (φ ∧ ψ) ∧ γ
∧-assoc1 = ∧-I (∧-I (∧-E₁ (axiom Z))
                    (∧-E₁ (∧-E₂ (axiom Z))))
               (∧-E₂ (∧-E₂ (axiom Z)))
```

The proof of the second entailment in Agda:

```
∧-assoc2 : ∀ {φ ψ γ : Formula} → Γ : (φ ∧ ψ) ∧ γ ⊢ φ ∧ (ψ ∧ γ)
∧-assoc2 = ∧-I (∧-E₁ (∧-E₁ (axiom Z)))
               (∧-I (∧-E₂ (∧-E₁ (axiom Z)))
                    (∧-E₂ (axiom Z)))
```

Therefore, we can conclude that the following equivalence holds:

```
ass-eq-∧ : ∀ {φ ψ γ : Formula} → φ ∧ (ψ ∧ γ) ~ (φ ∧ ψ) ∧ γ
ass-eq-∧ = ∧-assoc1 , ∧-assoc2
```

□

Proposition 4. $\phi \vee (\psi \vee \gamma) \sim (\phi \vee \psi) \vee \gamma$

Proof. We need to show $\Gamma, \phi \vee (\psi \vee \gamma) \vdash (\phi \vee \psi) \vee \gamma$ and $\Gamma, (\phi \vee \psi) \vee \gamma \vdash \phi \wedge (\psi \vee \gamma)$ Proof is done through natural deduction. The proofs are very similar and we have therefore opted to omitt the second proof, and we refer the reader to A.2 for details on the first one. In Agda we write the proofs as follows, the first entailment:

```
∨-assoc1 : ∀ {φ ψ γ : Formula} → Γ : φ ∨ (ψ ∨ γ) ⊢ (φ ∨ ψ) ∨ γ
∨-assoc1 = ∨-E (axiom Z)
                (∨-I₂ (∨-I₂ (axiom Z)))
                (∨-E (axiom Z)
                     (∨-I₂ (∨-I₁ (axiom Z)))
                     (∨-I₁ (axiom Z)))
```

The proof of the second entailment in `Agda`:

```
∨-assoc2 : ∀ {φ ψ γ : Formula} → Γ : (φ ∨ ψ) ∨ γ ⊢ φ ∨ (ψ ∨ γ)
∨-assoc2 = ∨-E (axiom Z)
                (∨-E (axiom Z)
                     (∨-I₂ (axiom Z))
                     (∨-I₁ (∨-I₂ (axiom Z))))
                (∨-I₁ (∨-I₁ (axiom Z)))
```

With these two results we have proven $\phi \vee (\psi \vee \gamma) \sim (\phi \vee \psi) \vee \gamma$. In `Agda`, the equivalence is shown as follows:

```
ass-eq-∨ : ∀ {φ ψ γ : Formula} → φ ∨ (ψ ∨ γ) ∼ (φ ∨ ψ) ∨ γ
ass-eq-∨ = ∨-assoc1 , ∨-assoc2
```

□

3.4.4 Distributivity

Proposition 5. $\phi \wedge (\psi \vee \gamma) \sim (\phi \wedge \psi) \vee (\phi \wedge \gamma)$

Proof. We need to show $\Gamma, \phi \wedge (\psi \vee \gamma) \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)$ and $\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \phi \wedge (\psi \vee \gamma)$. For the natural deductions, please refer to A.3 and A.4, together they show $\phi \wedge (\psi \vee \gamma) \sim (\phi \wedge \psi) \vee (\phi \wedge \gamma)$. The corresponding Agda proof of the first entailment is:

```
∧-dist1 : ∀ {φ ψ γ : Formula} → Γ : φ ∧ (ψ ∨ γ) ⊢ (φ ∧ ψ) ∨ (φ ∧ γ)
∧-dist1 = ∨-E (∧-E₂ (axiom Z))
              (∨-I₂ (∧-I (weakening (∧-E₁ (axiom Z))) (axiom Z)))
              (∨-I₁ (∧-I (weakening (∧-E₁ (axiom Z))) (axiom Z)))
```

In `Agda`, the second entailment is formalised as follows:

```
∧-dist2 : ∀ {φ ψ γ : Formula} → Γ : (φ ∧ ψ) ∨ (φ ∧ γ) ⊢ φ ∧ (ψ ∨ γ)
∧-dist2 = ∧-I (∨-E (axiom Z)
                   (∧-E₁ (axiom Z))
                   (∧-E₁ (axiom Z)))
              (∨-E (axiom Z)
                   (∨-I₂ (∧-E₂ (axiom Z)))
                   (∨-I₁ (∧-E₂ (axiom Z))))
```

We can combine these two proofs to show equivalence in `Agda`:

```
dist-eq-∧ : ∀ {φ ψ γ : Formula} → φ ∧ (ψ ∨ γ) ∼ (φ ∧ ψ) ∨ (φ ∧ γ)
dist-eq-∧ = ∧-dist1 , ∧-dist2
```

□

Proposition 6. $\phi \vee (\psi \wedge \gamma) \sim (\phi \vee \psi) \wedge (\phi \vee \gamma)$

Proof. We aim to show $\Gamma, \phi \vee (\psi \wedge \gamma) \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)$ and $\Gamma, (\phi \vee \psi) \wedge (\phi \vee \gamma) \vdash \phi \vee (\psi \wedge \gamma)$. For the deduction details please refer to A.5 and A.6. The following is the proof of the first entailment in Agda:

```
∨-dist1 : ∀ {φ ψ γ : Formula} → Γ : φ ∨ (ψ ∧ γ) ⊢ (φ ∨ ψ) ∧ (φ ∨ γ)
∨-dist1 = ∨-E (axiom Z)
              (∧-I (∨-I₂ (axiom Z))
                   (∨-I₂ (axiom Z)))
              (∧-I (∨-I₁ (∧-E₁ (axiom Z)))
                   (∨-I₁ (∧-E₂ (axiom Z))))
```

The second entailment is shown in Agda as follows:

```
∨-dist2 : ∀ {φ ψ γ : Formula} → Γ : (φ ∨ ψ) ∧ (φ ∨ γ) ⊢ φ ∨ (ψ ∧ γ)
∨-dist2 = ∨-E (∧-E₁ (axiom Z))
              (∨-I₂ (axiom Z))
              (∨-E (∧-E₂ (weakening (axiom Z)))
                   (∨-I₂ (axiom Z))
                   (∨-I₁ (∧-I (weakening (axiom Z)) (axiom Z))))
```

We can thus conclude that the two statements are equivalent:

```
dist-eq-∨ : ∀ {φ ψ γ : Formula} → φ ∨ (ψ ∧ γ) ∼ (φ ∨ ψ) ∧ (φ ∨ γ)
dist-eq-∨ = ∨-dist1 , ∨-dist2
```

□

4 Lindenbaum-Tarski algebra in Cubical Agda

4.1 Definition

Definition 4.1 (Lindenbaum-Tarski algebra)**.** *The Lindenbaum-Tarski algebra is the quotient algebra obtained by factoring the algebra of formulas by the equivalence relation* $\sim$*.*

We use a function from the `Cubical Agda` library to define the Lindenbaum-Tarski algebra.

```
LindenbaumTarski : Type
LindenbaumTarski = Formula / _∼_
```

4.2 Binary operations and propositional constants

4.3 Proof that the Lindenbaum Tarski algebra is Boolean

4.4 Soundness

References

[1] Agda documentation, 2023. URL: https://agda.readthedocs.io/en/v2.6.3/index.html.

[2] Ana Bove and Peter Dybjer. Dependent types at work, 2008. URL: https://www.cse.chalmers.se/~peterd/papers/DependentTypesAtWork.pdf.

[3] Dick van Dalen. *Logic and structure*. Springer, fifth edition, 2013.

A Derivations

A.1 Derivation of $\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash (\phi \wedge \psi) \wedge \gamma$

$$\dfrac{\mathcal{D}_1 \qquad \mathcal{D}_2}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash (\phi \wedge \psi) \wedge \gamma}\ \wedge\text{-I}$$

$\mathcal{D}_1$

$$\dfrac{\dfrac{\dfrac{\dfrac{\phi \wedge (\psi \wedge \gamma) \in \Gamma, \phi \wedge (\psi \wedge \gamma)}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \phi \wedge (\psi \wedge \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \psi \wedge \gamma}\ \wedge\text{-E}_2}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \psi}\ \wedge\text{-E}_1 \qquad \dfrac{\dfrac{\phi \wedge (\psi \wedge \gamma) \in \Gamma'}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \phi \wedge (\psi \wedge \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \phi}\ \wedge\text{-E}_1}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \phi \wedge \psi}\ \wedge\text{-I}$$

$\mathcal{D}_2$

$$\dfrac{\dfrac{\dfrac{\phi \wedge (\psi \wedge \gamma) \in \Gamma'}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \phi \wedge (\psi \wedge \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \psi \wedge \gamma}\ \wedge\text{-E}_2}{\Gamma, \phi \wedge (\psi \wedge \gamma) \vdash \gamma}\ \wedge\text{-E}_2$$

A.2 Derivation of $\Gamma, \phi \vee (\psi \vee \gamma) \vdash (\phi \vee \psi) \vee \gamma$

$$
\dfrac{
\dfrac{\phi \vee (\psi \vee \gamma) \in \Gamma, \phi \vee (\psi \vee \gamma)}{\Gamma, \phi \vee (\psi \vee \gamma) \vdash \phi \vee (\psi \vee \gamma)}\,\textsc{axiom}
\qquad
\dfrac{\dfrac{\dfrac{\phi \in \Gamma, \phi \vee (\psi \vee \gamma), \phi}{\Gamma, \phi \vee (\psi \vee \gamma), \phi \vdash \phi}\,\textsc{axiom}}{\Gamma, \phi \vee (\psi \vee \gamma), \phi \vdash \phi \vee \psi}\,\vee\text{-I}_2}{\Gamma, \phi \vee (\psi \vee \gamma), \phi \vdash (\phi \vee \psi) \vee \gamma}\,\vee\text{-I}_2
\qquad
\begin{matrix}\mathcal{D} \\ \Gamma, \phi \vee (\psi \vee \gamma), (\psi \vee \gamma) \vdash \psi \vee \gamma\end{matrix}
}{\Gamma, \phi \vee (\psi \vee \gamma) \vdash (\phi \vee \psi) \vee \gamma}\,\vee\text{-E}
$$

$\mathcal{D}$

$$
\dfrac{
\dfrac{\psi \vee \gamma \in \Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma \vdash \psi \vee \gamma}\,\textsc{axiom}
\qquad
\begin{matrix}\mathcal{D}' \\ \Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \psi \vdash (\phi \vee \psi) \vee \gamma\end{matrix}
\qquad
\dfrac{\dfrac{\gamma \in \Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \gamma}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \gamma \vdash \gamma}\,\textsc{axiom}}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \gamma \vdash (\phi \vee \psi) \vee \gamma}\,\vee\text{-I}_1
}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma \vdash (\phi \vee \psi) \vee \gamma}\,\vee\text{-E}
$$

$\mathcal{D}'$

$$
\dfrac{\dfrac{\dfrac{\psi \in \Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \psi}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \psi \vdash \psi}\,\textsc{axiom}}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \psi \vdash \phi \vee \psi}\,\vee\text{-I}_1}{\Gamma, \phi \vee (\psi \vee \gamma), \psi \vee \gamma, \psi \vdash (\phi \vee \psi) \vee \gamma}\,\vee\text{-I}_2
$$

A.3 Derivation of $\Gamma, \phi \wedge (\psi \vee \gamma) \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)$

$$
\dfrac{
\dfrac{\dfrac{\phi \wedge (\psi \vee \gamma) \in \Gamma, \phi \wedge (\psi \vee \gamma)}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \phi \wedge (\psi \vee \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \psi \vee \gamma}\ \wedge\text{-E}_2
\qquad
\begin{array}{c}\mathcal{D}_1\\ \Gamma, \phi \wedge (\psi \vee \gamma), \psi \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)\end{array}
\qquad
\begin{array}{c}\mathcal{D}_2\\ \Gamma, \phi \wedge (\psi \vee \gamma), \gamma \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)\end{array}
}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)}\ \vee\text{-E}
$$

$\mathcal{D}_1$

$$
\dfrac{\dfrac{
\dfrac{\dfrac{\dfrac{\phi \wedge (\psi \vee \gamma) \in \Gamma, \phi \wedge (\psi \vee \gamma)}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \phi \wedge (\psi \vee \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \phi}\ \wedge\text{-E}_1}{\Gamma, \phi \wedge (\psi \vee \gamma), \psi \vdash \phi}\ \textsc{weakening}
\qquad
\dfrac{\psi \in \Gamma, \phi \wedge (\psi \vee \gamma), \psi}{\Gamma, \phi \wedge (\psi \vee \gamma), \psi \vdash \psi}\ \textsc{axiom}
}{\Gamma, \phi \wedge (\psi \vee \gamma), \psi \vdash \phi \wedge \psi}\ \wedge\text{-I}}{\Gamma, \phi \wedge (\psi \vee \gamma), \psi \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)}\ \vee\text{-I}_1
$$

$\mathcal{D}_2$

$$
\dfrac{\dfrac{
\dfrac{\dfrac{\dfrac{\phi \wedge (\psi \vee \gamma) \in \Gamma'}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \phi \wedge (\psi \vee \gamma)}\ \textsc{axiom}}{\Gamma, \phi \wedge (\psi \vee \gamma) \vdash \phi}\ \wedge\text{-E}_1}{\Gamma, \phi \wedge (\psi \vee \gamma), \gamma \vdash \phi}\ \textsc{weakening}
\qquad
\dfrac{\gamma \in \Gamma, \phi \wedge (\psi \vee \gamma), \gamma}{\Gamma, \phi \wedge (\psi \vee \gamma), \gamma \vdash \gamma}\ \textsc{axiom}
}{\Gamma, \phi \wedge (\psi \vee \gamma), \gamma \vdash \phi \wedge \gamma}\ \wedge\text{-I}}{\Gamma, \phi \wedge (\psi \vee \gamma), \gamma \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)}\ \vee\text{-I}_1
$$

A.4 Derivation of $\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \phi \wedge (\psi \vee \gamma)$

$$\dfrac{\begin{matrix}\mathcal{D}_1 \\ \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \phi\end{matrix} \qquad \begin{matrix}\mathcal{D}_2 \\ \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \psi \vee \gamma\end{matrix}}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \phi \wedge (\psi \vee \gamma)}\ \wedge\text{-I}$$

$\mathcal{D}_1$

$$\dfrac{\dfrac{\dfrac{\phi \wedge \psi \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi \vdash \phi \wedge \psi}\ \textsc{axiom}}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi \vdash \phi}\ \wedge\text{-E}_1 \quad \dfrac{(\phi \wedge \psi) \vee (\phi \wedge \gamma) \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma)}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)}\ \textsc{axiom} \quad \dfrac{\dfrac{\phi \wedge \gamma \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma \vdash \phi \wedge \gamma}\ \textsc{axiom}}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma \vdash \phi}\ \wedge\text{-E}_1}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \phi}\ \vee\text{-E}$$

$\mathcal{D}_2$

$$\dfrac{\dfrac{\dfrac{\dfrac{\phi \wedge \psi \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi \vdash \phi \wedge \psi}\ \textsc{axiom}}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi \vdash \psi}\ \wedge\text{-E}_2}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \psi \vdash \psi \vee \gamma}\ \vee\text{-I}_2 \quad \dfrac{(\phi \wedge \psi) \vee (\phi \wedge \gamma) \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma)}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash (\phi \wedge \psi) \vee (\phi \wedge \gamma)}\ \textsc{axiom} \quad \dfrac{\dfrac{\dfrac{\phi \wedge \gamma \in \Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma \vdash \phi \wedge \gamma}\ \textsc{axiom}}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma \vdash \gamma}\ \wedge\text{-E}_2}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma), \phi \wedge \gamma \vdash \psi \vee \gamma}\ \vee\text{-I}_1}{\Gamma, (\phi \wedge \psi) \vee (\phi \wedge \gamma) \vdash \psi \vee \gamma}\ \vee\text{-E}$$

A.5 Derivation of $\Gamma, \phi \vee (\psi \wedge \gamma) \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)$

$$\cfrac{\cfrac{\phi \vee (\psi \wedge \gamma) \in \Gamma, \phi \vee (\psi \wedge \gamma)}{\Gamma, \phi \vee (\psi \wedge \gamma) \vdash \phi \vee (\psi \wedge \gamma)}\ \textsc{axiom} \qquad \begin{matrix}\mathcal{D}_1 \\ \Gamma, \phi \vee (\psi \wedge \gamma) \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)\end{matrix} \qquad \begin{matrix}\mathcal{D}_2 \\ \Gamma, \phi \vee (\psi \wedge \gamma) \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)\end{matrix}}{\Gamma, \phi \vee (\psi \wedge \gamma) \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \vee\text{-E}$$

$\mathcal{D}_1$

$$\cfrac{\cfrac{\cfrac{\phi \in \Gamma, \phi \vee (\psi \wedge \gamma), \phi}{\Gamma, \phi \vee (\psi \wedge \gamma), \phi \vdash \phi}\ \textsc{axiom}}{\Gamma, \phi \vee (\psi \wedge \gamma), \phi \vdash \phi \vee \psi}\ \vee\text{-I}_2 \qquad \cfrac{\cfrac{\phi \in \Gamma, \phi \vee (\psi \wedge \gamma), \phi}{\Gamma, \phi \vee (\psi \wedge \gamma), \phi \vdash \phi}\ \textsc{axiom}}{\Gamma, \phi \vee (\psi \wedge \gamma), \phi \vdash (\phi \vee \gamma)}\ \vee\text{-I}_2}{\Gamma, \phi \vee (\psi \wedge \gamma), \phi \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \wedge\text{-I}$$

$\mathcal{D}_2$

$$\cfrac{\cfrac{\cfrac{\cfrac{\psi \wedge \gamma \in \Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash \psi \wedge \gamma}\ \textsc{axiom}}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash \psi}\ \wedge\text{-E}_1}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash (\phi \vee \psi)}\ \vee\text{-I}_1 \qquad \cfrac{\cfrac{\cfrac{\psi \wedge \gamma \in \Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash \psi \wedge \gamma}\ \textsc{axiom}}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash \gamma}\ \wedge\text{-E}_2}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash (\phi \vee \gamma)}\ \vee\text{-I}_1}{\Gamma, \phi \vee (\psi \wedge \gamma), \psi \wedge \gamma \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \wedge\text{-I}$$

A.6 Derivation of $\Gamma, (\phi \vee \psi) \wedge (\phi \vee \gamma), \phi \vdash \phi \vee (\psi \wedge \gamma)$

Denote $\Gamma' = \Gamma \cup \{(\phi \vee \psi) \wedge (\phi \vee \gamma)\}$

$$\dfrac{\dfrac{\dfrac{(\phi \vee \psi) \wedge (\phi \vee \gamma) \in \Gamma'}{\Gamma' \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \textsc{axiom}}{\Gamma' \vdash \phi \vee \psi}\ \wedge\text{-E}_1 \qquad \dfrac{\dfrac{\phi \in \Gamma', \phi}{\Gamma', \phi \vdash \phi}\ \textsc{axiom}}{\Gamma', \phi \vdash \phi \vee (\psi \wedge \gamma)}\ \vee\text{-I}_2 \qquad \begin{matrix}\mathcal{D} \\ \Gamma', \psi \vdash \phi \vee (\psi \wedge \gamma)\end{matrix}}{\Gamma', \phi \vdash \phi \vee (\psi \wedge \gamma)}\ \vee\text{-E}$$

$\mathcal{D}$

$$\dfrac{\dfrac{\dfrac{\dfrac{(\phi \vee \psi) \wedge (\phi \vee \gamma) \in \Gamma'}{\Gamma' \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \textsc{axiom}}{\Gamma', \psi \vdash (\phi \vee \psi) \wedge (\phi \vee \gamma)}\ \textsc{weakening}}{\Gamma', \psi \vdash \phi \vee \gamma}\ \wedge\text{-E}_2 \qquad \dfrac{\dfrac{\phi \in \Gamma', \psi, \phi}{\Gamma', \psi, \phi \vdash \phi}\ \textsc{axiom}}{\Gamma', \psi, \phi \vdash \phi \vee (\psi \wedge \gamma)}\ \vee\text{-I}_2 \qquad \mathcal{D}'}{\Gamma', \psi, \phi \vdash \phi \vee (\psi \wedge \gamma)}\ \vee\text{-E}$$

$\mathcal{D}'$

$$\dfrac{\dfrac{\dfrac{\dfrac{\psi \in \Gamma', \psi}{\Gamma', \psi \vdash \psi}\ \textsc{axiom}}{\Gamma', \psi, \gamma \vdash \psi}\ \textsc{weakening} \qquad \dfrac{\gamma \in \Gamma', \psi, \gamma}{\Gamma', \psi, \gamma \vdash \gamma}\ \textsc{axiom}}{\Gamma', \psi, \gamma \vdash \psi \wedge \gamma}\ \wedge\text{-I}}{\Gamma', \psi, \gamma \vdash \phi \vee (\psi \wedge \gamma)}\ \vee\text{-I}_1$$

B LindenbaumTarski.agda